


THE ANTENNA LABORATORY

RESEARCH ACTIVITIES in ---

Automatic Controls Antennas Echo Area Studies
Microwave Circuits Astronautics E M Field Theory
Terrain Investigations Radomes System Analysis
Wave Propagation Submillimeter Applications



EFFICIENT RECURSIVE SOLUTIONS FOR PLANE AND CYLINDRICAL MULTILAYERS

J. H. Richmond

Contract Number NOw 65-0329-d

1968-1

10 August 1965

Prepared for:
Bureau of Naval Weapons
Washington, D. C.

Department of ELECTRICAL ENGINEERING



CLEARINGHOUSE FOR FEDERAL SCIENTIFIC AND TECHNICAL INFORMATION			
Hardcopy	11/11/1965		
\$2.00	\$0.50	27 pp	22
REPRODUCTION COPY			
Code 1			

THE OHIO STATE UNIVERSITY
RESEARCH FOUNDATION
Columbus, Ohio

REPORT 1968-1

REPORT

by

THE OHIO STATE UNIVERSITY RESEARCH FOUNDATION
COLUMBUS, OHIO 43212

Sponsor	Bureau of Naval Weapons Washington, D. C.
Contract Number	NOw 65-0329-d
Investigation of	Reactive-Wall Reflectors for 9 mm Microwave Operation
Subject of Report	Efficient Recursive Solutions for Plane and Cylindrical Multilayers
Submitted by	J. H. Richmond Antenna Laboratory Department of Electrical Engineering
Date	10 August 1965

Release to Defense Center (ASTIA) and to OTS, Dept. of Commerce, Washington, D. C., without restriction. The contractor imposes no limitation or restriction upon the distribution of this document by the Government to both Government and non-Government recipients.

ABSTRACT

When a plane harmonic wave has oblique incidence on a stack of homogeneous dielectric slabs, the field in each layer consists of an outgoing and a reflected wave. The complex amplitudes of these plane waves in layer n are denoted by A_n and B_n , and simple recursive relations are derived for A_{n+1} and B_{n+1} as functions of A_n and B_n . The solution begins by setting $A_0 = 1$ and $B_0 = 0$, and it is completed by calculating the transmission and reflection coefficients of the plane multilayer as simple functions of A_{N+1} and B_{N+1} where N represents the total number of layers.

This technique is also applicable when a plane wave is incident on a multilayer dielectric cylinder or sphere. The required equations are derived for the cylindrical case with normal incidence. The rigorous solutions and digital-computer programs are included for both principal polarizations for plane and cylindrical multilayers.

TABLE OF CONTENTS

	Page
I. INTRODUCTION	1
II. THE THEORY OF THE PLANE MULTILAYER	1
III. THE THEORY OF THE CYLINDRICAL MULTILAYER	6
IV. NUMERICAL RESULTS	11
V. CONCLUSIONS	11
REFERENCE	12
APPENDIX I - COMPUTER PROGRAM FOR THE PLANE MULTILAYER	15
APPENDIX II - COMPUTER PROGRAM FOR THE CYLINDRICAL MULTILAYER	19

EFFICIENT RECURSIVE SOLUTIONS FOR PLANE AND CYLINDRICAL MULTILAYERS

I. INTRODUCTION

The transmission and reflection coefficients of a plane dielectric multilayer are pertinent in the design of radomes and in other applications. One type of corner reflector has one of its three conducting surfaces coated with several dielectric layers which are designed to produce a right-circularly polarized reflection when the incident wave is right-circularly polarized. Multilayer dielectric spheres and cylinders form an important class of antenna scanning and echo enhancement devices.

A matrix-multiplication solution for the plane multilayer is described by Collin[1]. This matrix solution represents a distinct advance over the older methods. A recursive formulation developed here is even more straightforward in its theory and more efficient for numerical calculations. The technique is applicable also to radially layered cylinders and spheres, waveguides containing two or more media, and surface waves on plane multilayers.

This report presents the derivation of the recursion formulas for plane and cylindrical multilayers. The two principal polarizations are considered, and digital-computer programs are included. This solution can be applied directly to the analyses of inhomogeneous slabs and cylinders by using piecewise uniform approximations for the permeability and permittivity functions.

II. THE THEORY OF THE PLANE MULTILAYER

Suppose a harmonic plane wave in free space has oblique incidence on a plane multilayer consisting of N homogeneous isotropic slabs, as indicated in Fig. 1. Let d_n , μ_n and ϵ_n represent the thickness, permeability, and permittivity of slab n . The slabs are considered to have infinite width and height and parallel surfaces, with unbounded free space on both sides of the multilayer. The incident plane wave impinging on the left-hand surface of the multilayer is given, in the TE case (i. e. perpendicular polarization) by




Fig. 1. A plane multilayer, illustrating the outgoing and reflected waves in each layer.

$$(1) \quad E_x^i = E_0 e^{jk_0 y \sin \theta} e^{jk_0 z \cos \theta},$$

where θ is the angle of incidence, $k_0 = 2\pi/\lambda$, and λ is the free-space wavelength. The reflected plane wave is given by

$$(2) \quad E_x^r = R E_0 e^{jk_0 y \sin \theta} e^{-jk_0 z \cos \theta},$$

where R is the reflection coefficient of the multilayer. The transmitted plane wave on the right-hand side of the multilayer is represented by

$$(3) \quad E_x^t = T E_0 e^{jk_0 y \sin \theta} e^{jk_0 z \cos \theta},$$

where T is the transmission coefficient of the multilayer. The field in each layer can be regarded as an infinite series of plane waves bouncing back and forth, but it is more convenient (and equally valid) to consider it to be the sum of only two plane waves, one traveling outward and one reflected.

In layer n, for example, the field is represented by

$$(4) \quad E_n = \left(A_n e^{\gamma_n z} + B_n e^{-\gamma_n z} \right) e^{jk_0 y \sin \theta} .$$

Similarly, in layer n+1 the electric field intensity is given by

$$(5) \quad E_{n+1} = \left(A_{n+1} e^{\gamma_{n+1} z} + B_{n+1} e^{-\gamma_{n+1} z} \right) e^{jk_0 y \sin \theta} .$$

The boundary between layers n and n+1 is located at

$$(6) \quad z_n = d_1 + d_2 + d_3 + \dots + d_n .$$

By enforcing the boundary conditions on E_x and H_y at $z = z_n$, we can show that

$$(7) \quad A_{n+1} = P_n A_n + Q_n B_n$$

and

$$(8) \quad B_{n+1} = R_n A_n + S_n B_n ,$$

where

$$(9) \quad P_n = 0.5 (1 + \dot{\mu}_{n+1} \gamma_n / \dot{\mu}_n \gamma_{n+1}) e^{(\gamma_n - \gamma_{n+1}) z_n} ,$$

$$(10) \quad Q_n = 0.5 (1 - \dot{\mu}_{n+1} \gamma_n / \dot{\mu}_n \gamma_{n+1}) e^{-(\gamma_n + \gamma_{n+1}) z_n} ,$$

$$(11) \quad R_n = 0.5 (1 - \dot{\mu}_{n+1} \gamma_n / \dot{\mu}_n \gamma_{n+1}) e^{(\gamma_n + \gamma_{n+1}) z_n} ,$$

and

$$(12) \quad S_n = 0.5 (1 + \dot{\mu}_{n+1} \gamma_n / \dot{\mu}_n \gamma_{n+1}) e^{-(\gamma_n - \gamma_{n+1}) z_n} .$$

The propagation constant γ_n for layer n will be complex if the medium is dissipative. Both the real and imaginary parts of γ_n will be positive. If layer n is a lossless medium, γ_n will be purely imaginary. The wave equation is employed to obtain the following equation for γ_n :

$$(13) \quad \gamma_n = j \sqrt{\omega^2 \mu_n \epsilon_n - k_0^2 \sin^2 \theta} \quad .$$

The reflection and transmission coefficients of the multilayer (R and T) are often of particular interest. We can calculate these quantities in a systematic manner by setting

$$(14) \quad A_0 = 1$$

and

$$(15) \quad B_0 = 0$$

and then using the recursion equations Eqs. (7) and (8) to calculate A_1 , B_1 , A_2 , B_2 , ... A_{N+1} , and B_{N+1} in that order. This process is usually carried out on a digital computer.

From Eqs. (1) through (4),

$$(16) \quad E_0 = A_{N+1} \quad ,$$

$$(17) \quad R = \frac{B_{N+1}}{A_{N+1}} \quad ,$$

and

$$(18) \quad T = \frac{1}{A_{N+1}}$$

In the TE case the constants A_n and B_n represent the electric field intensities of the outgoing and reflected waves in each layer. In the TM case (parallel polarization) the solution proceeds in the same manner. In fact, the equations given above apply in both cases but the A_n and B_n represent the magnetic field intensities in the TM case and μ_{n+1} and μ_n must be replaced with ϵ_{n+1} and ϵ_n in Eqs. (9) through (12).

If a perfectly conducting sheet is placed on the right-hand surface of the multilayer (i. e., on the xy plane), the solution is again given by the equations above with the exception that we do not calculate the transmission coefficient T in this case, and Eqs. (14) and (15) are replaced with

$$(19) \quad A_1 = 1 \text{ and } B_1 = -1 \text{ in the TE case}$$

and

$$(20) \quad A_1 = 1 \text{ and } B_1 = 1 \text{ in the TM case.}$$

Equations (19) and (20) are obtained by forcing the tangential electric field intensity to vanish at the perfectly conducting plane.

The insertion phase delay, denoted by the symbol IPD, is equal to the phase angle of the complex transmission coefficient T but is of opposite sign:

$$(21) \quad \text{IPD} = - \text{Phase} (T)$$

In the previous equations, the reflection coefficient R is defined as the ratio of the reflected wave amplitude to the incident wave amplitude at the coordinate origin; that is,

$$(22) \quad R = \frac{E_x^r(0, 0, 0)}{E_x^i(0, 0, 0)} \text{ for the TE case}$$

and

$$(23) \quad R = \frac{H_x^r(0, 0, 0)}{H_x^i(0, 0, 0)} \text{ for the TM case.}$$

Let R' represent the ratio of the reflected and incident wave amplitudes at the point of incidence where $(x, y, z) = (0, 0, d)$, with d representing the total thickness of the multilayer. Then

$$(24) \quad R' = R e^{-2jk_0 d \cos \theta}.$$

Equation (24) can be verified by means of Eqs. (1) and (2) and the definition of R' . The two reflection coefficients, R and R' , differ only in phase.

A digital-computer program based on these equations is given in Appendix I, and some numerical results are included in Section IV.

III. THE THEORY OF THE CYLINDRICAL MULTILAYER

Consider a plane harmonic wave in free space to have normal incidence on a dielectric cylinder of infinite length, as suggested in Fig. 2. The cylinder axis is taken to be the z axis in a rectangular coordinate system, and the x axis is the axis of propagation of the incident plane wave. Let the cylinder consist of M lossless homogeneous layers, each




Fig. 2. Circular dielectric cylinder with several homogeneous layers.

layer being a circular cylindrical shell. The permeability and permittivity of layer m are denoted by μ_m and ϵ_m , and the phase constant is given by

$$(25) \quad k_m = \omega \sqrt{\mu_m \epsilon_m} .$$

In the TM case the electric field intensity has only a z component, given in layer m by

$$(26) \quad E_m = \sum_{n=0}^{\infty} [A_{mn} J_n(k_m \rho) + B_{mn} N_n(k_m \rho)] \cos n\phi ,$$

where (ρ, ϕ, z) are the cylindrical coordinates and J_n and N_n represent the Bessel and Neumann functions. The time dependence, $e^{j\omega t}$, is understood. In the TE case the solution is obtained from the equations given here by interchanging μ and ϵ and $\underline{E}$ and $\underline{H}$, where $\underline{H}$ represents the magnetic field intensity.

The coefficients A_{mn} and B_{mn} must be determined by applying the boundary conditions on E_z and H_ϕ at each interface. From Maxwell's equations, the ϕ component of the magnetic field intensity in layer m is given by

$$(27) \quad H_m = (k_m / j\omega\mu_m) \sum_{n=0}^{\infty} [A_{mn} J'_n(k_m \rho) + B_{mn} N'_n(k_m \rho)] \cos n\phi ,$$

where J'_n and N'_n denote the derivatives of the Bessel and Neumann functions with respect to the argument.

Let R_m denote the outer radius of layer m . From the boundary conditions at this interface between layers m and $m+1$, it is found that the wave expansion coefficients for the two regions have the following linear relations:

$$(28) \quad A_{m+1,n} = U_{mn} A_{mn} + W_{mn} B_{mn}$$

and

$$(29) \quad B_{m+1,n} = V_{mn} A_{mn} + X_{mn} B_{mn} ,$$

where

$$(30) \quad U_{mn} = (\pi R_m / 2\mu_m) [-\mu_{m+1} k_m J'_n(k_m R_m) N_n(k_{m+1} R_m) \\ + \mu_m k_{m+1} J_n(k_m R_m) N'_n(k_{m+1} R_m)] ,$$

$$(31) \quad V_{mn} = (\pi R_m / 2\mu_m) [\mu_{m+1} k_m J'_n(k_m R_m) J_n(k_{m+1} R_m) \\ - \mu_m k_{m+1} J_n(k_m R_m) J'_n(k_{m+1} R_m)] ,$$

$$(32) \quad W_{mn} = (\pi R_m / 2\mu_m) [-\mu_{m+1} k_m N'_n(k_m R_m) N_n(k_{m+1} R_m) \\ + \mu_m k_{m+1} N_n(k_m R_m) N'_n(k_{m+1} R_m)] ,$$

and

$$(33) \quad X_{mn} = (\pi R_m / 2\mu_m) [\mu_{m+1} k_m N'_n(k_m R_m) J_n(k_{m+1} R_m) \\ - \mu_m k_{m+1} N_n(k_m R_m) J'_n(k_{m+1} R_m)] .$$

If the coefficients in the first layer (A_{1n} and B_{1n}) were known, the coefficients in the remaining layers could be calculated by using Eqs. (28) and (29) recursively. To permit a procedure of this type, let us define a set of normalized coefficients A'_{mn} and B'_{mn} related to A_{mn} and B_{mn} by the proportionality constant K_n as follows:

$$(34) \quad A_{mn} = K_n A'_{mn}$$

and

$$(35) \quad B_{mn} = K_n B'_{mn} .$$

With no loss of generality, let

$$(36) \quad A'_{1n} = 1 .$$

If the center layer is a dielectric medium, the field must be finite at the origin and

$$(37) \quad B'_{1n} = 0.$$

The normalized coefficients also obey the same recursion formulas and we can now calculate A'_{2n} , B'_{2n} , ... A'_{Mn} , B'_{Mn} , $A'_{M+1,n}$, and $B'_{M+1,n}$ in that order.

The field in the exterior free-space region is given by

$$(38) \quad E_{M+1} = \sum_{n=0}^{\infty} [(-j)^n e_n J_n(k_0 \rho) + C_n H_n^{(2)}(k_0 \rho)] \cos n\phi,$$

where $H_n^{(2)}(k_0 \rho)$ represents the Hankel function and k_0 is the phase constant of free space. The first series in Eq. (38) represents the incident plane-wave field, and the second series is the scattered field which contains outward-traveling waves only. The function e_n is unity if $n = 0$, and $e_n = 2$ if n is greater than zero.

The field in the exterior free-space region is also given by Eq. (26) with $m = M+1$. Comparison of the two representations reveals that

$$(39) \quad A_{M+1,n} - j B_{M+1,n} = (-j)^n e_n.$$

The constant of proportionality, K_n , is found from Eqs. (34), (35), and (39) to be

$$(40) \quad K_n = \frac{(-j)^n e_n}{A'_{M+1,n} - j B'_{M+1,n}}.$$

The scattering coefficients for the external region are given by

$$(41) \quad C_n = j B_{M+1,n} = \frac{-(-j)^n e_n B'_{M+1,n}}{B'_{M+1,n} + j A'_{M+1,n}}.$$

This completes the solution. Equation (26) can be employed to calculate the field at any point in the dielectric cylinder, and Eqs. (38) and (41) are used to calculate the field at any point outside the cylinder.

If the dielectric media are lossless, it may be noted that the constants A'_{mn} , B'_{mn} , U_{mn} , V_{mn} , W_{mn} , and X_{mn} are real. The external scattering coefficients, C_n , are complex.

In the case of a perfectly conducting circular cylinder with one or more dielectric layers, let "a" be the radius of the conducting cylinder and R_1 the outer radius of the first dielectric layer. The above equations again give the solution if Eq. (37) is replaced with

$$(42) \quad B'_{1n} = - \frac{J_n(k_1 a)}{N_n(k_1 a)} \quad \text{for the TM case,}$$

and

$$(43) \quad B'_{1n} = - \frac{J'_n(k_1 a)}{N'_n(k_1 a)} \quad \text{for the TE case.}$$

Equations (42) and (43) are obtained by setting $E_z = 0$ or $E_\phi = 0$ at the conducting surface.

At any point in space outside the dielectric cylinder, the scattered field is given in the TM case by

$$(44) \quad E^s = \sum_{n=0}^{\infty} (-j)^n e_n D_n H_n(k_0 \rho) \cos n\phi ,$$

where the superscript (2) is understood on the Hankel functions $H(k_0 \rho)$ and

$$(45) \quad D_n = \frac{-B'_{M+1, n}}{B'_{M+1, n} + j A'_{M+1, n}} .$$

In calculating the scattered field at a great distance from the cylinder, we use the asymptotic form for the Hankel functions of large argument to show that

$$(46) \quad E^s = \sqrt{2j/\pi k_0 \rho} e^{-jk_0 \rho} \sum_{n=0}^{\infty} e_n D_n \cos n\phi .$$

When a plane wave is incident on a cylindrical structure of infinite length, the distant scattering pattern is conveniently described by the echo width which is defined as follows:

$$(47) \quad W = \lim_{\rho \rightarrow \infty} 2\pi\rho \left| E^s/E^i \right|^2 .$$

In Eq. (38) the incident electric field intensity E^i is taken to have unit magnitude. From Eqs. (46) and (47) the bistatic echo width of the multilayer dielectric cylinder is given by

$$(48) \quad W = (2\lambda/\pi) \left| \sum_{n=0}^{\infty} e_n D_n \cos n\phi \right|^2 ,$$

where λ is the wavelength in free space.

A digital-computer program based on these equations is included in Appendix II.

IV. NUMERICAL RESULTS

Figure 3 shows the transmission coefficient and insertion phase delay versus frequency for a plane multilayer, calculated with the aid of the equations in Section II. Numerical calculations obtained with these equations have been found to agree with those published by other investigators.

Figure 4 shows the distant scattering pattern of a cylindrical multilayer, calculated with the equations given in Section III. These equations are also found to yield results which agree with previously published data.

V. CONCLUSIONS

An efficient recursive solution is developed for the transmission and reflection coefficients of a plane multilayer and for the scattering pattern of a cylindrical multilayer. The two principal polarizations are considered, and digital-computer programs are included. The technique is also applicable to the multilayered sphere.

The appropriate equations are also given for the reflection coefficients of a perfectly conducting plane which is coated with a stack of homogeneous dielectric sheets, and for the scattering pattern of a perfectly conducting circular cylinder coated with several homogeneous dielectric layers.

REFERENCE

1. Collin, R. E., Field Theory of Guided Waves, McGraw-Hill Book Co., New York (1960) pp. 79-85.




Fig. 3. Calculated power transmission coefficient and insertion phase delay versus frequency for plane multilayer having three layers.




Fig. 4. Distant scattering patterns of a five-layer dielectric cylinder, calculated with the recursion technique.

APPENDIX I

COMPUTER PROGRAM FOR THE PLANE MULTILAYER

A computer program for the reflection and transmission coefficients of a plane multilayer is shown in Fig. 5. This program, written in the computer language known as Scatran, is based on the equations given in Section II. The symbols used for the input data are defined as follows:

KKK	= Number of cases to be calculated with different frequencies or angles of incidence
N	= Number of layers
D(I)	= Thickness of layer i in inches
E(I)	= Dielectric constant of layer i relative to that of free space
TD(I)	= Electric loss tangent of layer i
U(I)	= Permeability of layer i relative to that of free space
TP(I)	= Magnetic loss tangent of layer i
THETA	= Angle of incidence in degrees
FGC	= Frequency in gigacycles

Some of the other symbols used in the program are defined below.

TH	= Angle of incidence in radians
WAVE	= Wavelength in free space, inches
SS	= $\sin^2 \theta$
CC	= $\cos \theta$
UC(I)	= Complex relative permeability of layer i
EC(I)	= Complex relative permittivity of layer i
G(I)	= γ/k_0 for layer i, as given by Eq. (13)
Z(I)	= $k_0 z_i$ where z_i is defined by Eq. (6)

The field intensities, A_n and B_n , of the outgoing and reflected waves in layer n are denoted by

AE and BE	for the TE case without a conducting plane,
AM and BM	for the TM case without a conducting plane,
AEC and BEC	for the TE case with conducting plane, and
AMC and BMC	for the TM case with conducting plane.

Between statements S30 and S50, the recursive calculations are carried out in accordance with Eqs. (7) through (12). Immediately following statement S50, the transmission and reflection coefficients are calculated by means of Eqs. (17), (18), and (24). These coefficients are denoted by

TE = Transmission coefficient for the TE case without
conducting plane,
 TM = Transmission coefficient for the TM case without
conducting plane,
 RE = Reflection coefficient R' for the TE case without
conducting plane,
 RM = Reflection coefficient R' for the TM case without
conducting plane,
 REC = Reflection coefficient R' for the TE case with a
conducting plane, and
 RMC = Reflection coefficient R' for the TM case with a
conducting plane.

The symbols used for the output data are defined by

REP, PRE, TEP, FIPDE = Power reflection coefficient,
reflection phase, power transmission
coefficient, and insertion phase delay
for TE case without conducting plane;
 RMP, PRM, TMP, FIPDM = Power reflection coefficient, reflec-
tion phase, power transmission co-
efficient, and insertion phase delay
for TM case without conducting plane;
 RECP, PREC = Power reflection coefficient and
reflection phase for TE case with
conducting plane; and
 RMCP, PRMC = Power reflection coefficient and
reflection phase for TM case with
conducting plane.

A typical set of input data in the proper format is shown at the end of the computer program.

This program will handle a maximum of 100 layers, but this number can be increased to a much larger number simply by modifying the dimension statement near the beginning of the program. Many of the statements can be deleted when the multilayer with a conducting plane is the only case of interest, or when this case is of no interest. Furthermore, the program can be simplified if all of the layers have the same permeability as free space. A simple modification in the program can provide for increments to be taken in the frequency or the angle of incidence.

```

***      RUN
***      SCATRAM
COMPLEX(CSQRTL.,CEXPL.,CI,UC,EC,G,AF,AM,BE,BM,AEC,BEC,GZ,
AMC,BMC,F,YI,YII,EXP1,EXP11,AFP,AMP,EXP,RF,RM,AFCP,BEC,
AMCP,RMC,FP,FM,TE,TM)-
DIMENSION(D(100),F(100),TD(100),U(100),TP(100),Z(100),
UC(100),EC(100),G(100))-
FIRST    READ INPUT,8,(KKK,N)-
          READ INPUT,7,((D(I),I=1,1,1,LE,N))-
          READ INPUT,7,((F(I),I=1,1,1,LE,N))-
          READ INPUT,7,((TD(I),I=1,1,1,LE,N))-
          READ INPUT,7,((U(I),I=1,1,1,LE,N))-
          READ INPUT,7,((TP(I),I=1,1,1,LE,N))-
          NN=N+1-
          DO THROUGH(S5),I=1,1,1,LE,N-
          FI=I-
S5        WRITE OUTPUT,2,(FI,D(I),F(I),TD(I),U(I),TP(I))-
          TP1=6.2831853-
          DR=57.295779-
          CI=.1,1.-
          EC(NN)=1.-
          UC(NN)=1.-
          Z(0)=.0-
          WRITE OUTPUT,2-
          DO THROUGH(S100),K=1,1,K,LE,KKK-
          READ INPUT,7,(THETA,FGC)-
          TH=.01745329*THETA-
          WAVE = 11.80314/FGC-
          WRITE OUTPUT,2,(THETA,FGC,WAVE)-
          WRITE OUTPUT,2-
          SS=SIN.(TH)*SIN.(TH)-
          CC=COS.(TH)-
          DO THROUGH(S20),I=1,1,1,LE,N-
          UC(I)=U(I)*(1.-CI*TP(I))-
          EC(I)=F(I)*(1.-CI*TD(I))-
S20        G(I)=CI*CSQRTL.(UC(I)*EC(I)-SS)-
          GZ=CI*CC-
          G(NN)=CI*CC-
          DO THROUGH(S30),I=1,1,1,LE,N-
          II=I-1-
S30        Z(I)=Z(II)+D(I)*TP1/WAVE-
          AE=.5*(1.+UC(I)*GZ/G(I))-
          BE=.5*(1.-UC(I)*GZ/G(I))-
          AM=.5*(1.+EC(I)*GZ/G(I))-
          BM=.5*(1.-EC(I)*GZ/G(I))-
          AFC=1.-
          BEC=-1.-
          AMC=1.-
          BMC=1.-
          DO THROUGH(S50),II=2,1,II,LE,NN-
          I=II-1-
          F=UC(II)*G(I)/(UC(I)*G(II))-
          YI=G(I)*Z(I)-
          YII=G(II)*Z(I)-

```

Fig. 5. Digital-computer program for reflection and transmission coefficients of plane multilayer.

```

FP=1.+F-
FM=1.-F-
EXP1=CFXPL.(Y1)-
EXP11=CFXPL.(Y11)-
AFP=AF-
AMP=AM-
AFCP=AEC-
AMCP=AMC-
AF=(.5/EXP11)*(AFP*EXP1*FP+BF*FM/EXP1)-
BF=.5*EXP11*(AFP*EXP1*FM+BF*FP/EXP1)-
AEC=(.5/EXP11)*(AFCP*EXP1*FP+BEC*FM/EXP1)-
BEC=.5*EXP11*(AFCP*EXP1*FM+BEC*FP/EXP1)-
F=EC(11)*G(1)/(EC(1)*C(11))-
FP=1.+F-
FM=1.-F-
AM=(.5/EXP11)*(AMP*EXP1*FP+BM*FM/EXP1)-
BM=.5*EXP11*(AMP*EXP1*FM+BM*FP/EXP1)-
AMC=(.5/EXP11)*(AMCP*EXP1*FP+BMC*FM/EXP1)-
BMC=.5*EXP11*(AMCP*EXP1*FM+BMC*FP/EXP1)-
TE=1./AE-
TM=1./AM-
ARG=2.*CC*Z(N)-
EXP=COS.(ARG)-.1.*SIN.(ARG)-
RF=EXP*RE/AE-
RM=EXP*RM/AM-
REC=EXP*BEC/AEC-
RMC=EXP*BMC/AMC-
TEP=(.ABS.TE).P.2-
TMP=(.ABS.TM).P.2-
REP=(.ABS.RE).P.2-
RMP=(.ABS.RM).P.2-
RECP=(.ABS.REC).P.2-
RMCP=(.ABS.RMC).P.2-
FIPDE=-DR*FATAN2.(.IMAG.TE,.REAL.TE)-
FIPDM=-DR*FATAN2.(.IMAG.TM,.REAL.TM)-
PRE=DR*FATAN2.(.IMAG.RE,.REAL.RE)-
PRM=DR*FATAN2.(.IMAG.RM,.REAL.RM)-
PREC=DR*FATAN2.(.IMAG.REC,.REAL.REC)-
PRMC=DR*FATAN2.(.IMAG.RMC,.REAL.RMC)-
WRITE OUTPUT,2,(REP,PRE,TEP,FIPDE)-
WRITE OUTPUT,2,(RMP,PRM,TMP,FIPDM)-
WRITE OUTPUT,2,(RECP,PREC)-
WRITE OUTPUT,2,(RMCP,PRMC)-
S100 WRITE OUTPUT,2-
END PROGRAM(FIRST)-
*** DATA
1 5
.15 .15 .15 .26 .26
5.0 4.0 3.0 2.0 1.5
.002 .002 .002 .002 .002
1.0 1.0 1.0 1.0 1.0
.0 .0 .0 .0 .0
.0 10.0

```

Fig. 5. Digital-computer program for reflection and transmission coefficients of plane multilayer. (cont.)

APPENDIX II

COMPUTER PROGRAM FOR THE CYLINDRICAL MULTILAYER

Figure 6 shows a computer program in Scatran for calculating the scattering coefficients and the scattering pattern of a layered dielectric cylinder, based on the equations developed in Section III. The input data symbols are defined as follows:

FN = Number of layers,
 DPH = Increment in scattering angle ϕ in degrees,
 RL(I) = Outer radius of layer i in free-space wavelengths,
 E(I) = Permittivity of layer i relative to that of free space, and
 U(I) = Permeability of layer i relative to that of free space.

This computer program is applicable only to lossless dielectric cylinders in which the center region is dielectric. With a slight modification it will give the solution appropriate for a perfectly-conducting center region. In the form shown, the program will handle a maximum of 100 layers, but this can be extended greatly by changing the dimension statement.

In this program the highest order of the Bessel and Neumann functions required is determined from the approximate equation $L = 4 + k_N R_N$, where N denotes the number of layers. In its present form, the program sets an upper limit of 100 for the maximum order of the Bessel and Neumann functions.

The wave amplitudes in the various layers are denoted by $A(I)$ and $B(I)$ for the TM case and $AP(I)$ and $BP(I)$ for the TE case, where I represents the mode number n . Between statements S25 and S26, the program sets $A_{1n} = 1$ and $B_{1n} = 0$ in accordance with Eqs. (36) and (37). Next, the arguments $k_m R_m$ and $k_{m+1} R_m$ for the Bessel functions are calculated and are denoted by W . Between statements S26 and S50, the Bessel and Neumann functions are obtained by calling a subroutine. The following notation is used:

FJA(I) = $J_i(k_m R_m)$,
 FJB(I) = $J_i(k_{m+1} R_m)$,
 FNA(I) = $N_i(k_m R_m)$, and
 FNB(I) = $N_i(k_{m+1} R_m)$.

In its present form, the subroutine for the Bessel and Neumann functions is limited to arguments smaller than 1000.

Between statements S50 and S80, the derivatives of the Bessel and Neumann functions are calculated. These are denoted by

$$\begin{aligned} \text{FJP} &= J'_j(k_m R_m), \\ \text{FJPP} &= J'_j(k_{m+1} R_m), \\ \text{FNP} &= N'_j(k_m R_m), \text{ and} \\ \text{FNPP} &= N'_j(k_{m+1} R_m). \end{aligned}$$

Between S80 and S100, the constants U_{mn} , V_{mn} , W_{mn} , and X_{mn} defined in Eqs. (30) through (33) are calculated and are denoted by UJ, VJ, WJ, and XJ for the TM case, and by UP, VP, WP, and XP for the TE case. Equations (28) and (29) are employed to calculate $A'_{m+1,n}$ and $B'_{m+1,n}$ which are denoted by A(J) and B(J) for the TM case and by AP(J) and BP(J) for the TE case. Between S110 and S150 the scattering coefficients D_n are calculated by means of Eq. (45). They are denoted by D(I) and DP(I) for the TM and TE cases, respectively.

In the remainder of the program Eq. (48) is used to calculate the echo width of the layered cylinder as a function of the angle ϕ . The symbols for the output data are defined by

$$\begin{aligned} \text{PHI} &= \text{Scattering angle } \phi \text{ in degrees,} \\ \text{ECHW} &= \text{Echo width/wavelength for TM case, and} \\ \text{ECHWP} &= \text{Echo width/wavelength for TE case.} \end{aligned}$$

```

***      RUN
***      SCATTRAN
      COMPLEX(EXP,ES,FF,FH,D,DP,ESP)-
      DIMENSION(RL(102),E(102),U(102),D(102),FJA(102),DP(102),
      FJB(102),FNA(102),FNB(102),A(102),B(102),AP(102),BP(102))-
FIRST    PI=3.1415926-
      TP=2.*PI-
      P2=PI/2.-
      CST=2./PI-
      READ INPUT,7,(FN,DPH)-
      N=FN-
      READ INPUT,7,((RL(I),I=1,1,1,LE,N))-
      READ INPUT,7,((E(I),I=1,1,1,LE,N))-
      READ INPUT,7,((U(I),I=1,1,1,LE,N))-
S25      DO THROUGH(S25),I=1,1,1,LE,N-
      WRITE OUTPUT,2,(RL(I),E(I),U(I))-
      WRITE OUTPUT,2-
      NN=N+1-
      L=4.+TP*RL(N)*SORT.(E(N)*U(N))-
      PROVIDED(L.G.100),L=100-
      LL=L-1-
      E(NN)=1.-
      U(NN)=1.-
      IJK=L-
      DO THROUGH(S26),I=0,1,1,LE,L-
      A(I)=1.-
      B(I)=.0-
      AP(I)=1.-
S26      BP(I)=.0-
      DO THROUGH(S110),M=1,1,M,LE,N-
      MM=M+1-
      L=IJK-
      W=TP*RL(M)*SORT.(U(M)*E(M))-
      CALL SUBROUTINE(FJA,FNA,L,IND)=BFX90.(W)-
      W=TP*RL(M)*SORT.(U(MM)*E(MM))-
      CALL SUBROUTINE(FJB,FNB,L,IND)=BFX90.(W)-
      LL=L+1-
      DO THROUGH(S50),IK=LL,1,IK,LE,IJK-
      FJA(IK)=.0-
      FNA(IK)=.0-
      FJB(IK)=.0-
S50      FNB(IK)=.0-
      CNST=P2*U(MM)*TP*RL(M)-
      CNSTP=P2*E(MM)*TP*RL(M)-
      YM=SORT.(E(M)/U(M))*CNST-
      YMP=SORT.(U(M)/E(M))*CNSTP-
      YMM=SORT.(E(MM)/U(MM))*CNST-
      YMMP=SORT.(U(MM)/E(MM))*CNSTP-
      DO THROUGH(S100),J=0,1,J,LE,LL-
      JJ=J+1-
      FJP=-FJA(JJ)-
      FJPP=-FJB(JJ)-
      FNP=-FNA(JJ)-
      FNPP=-FNB(JJ)-
      PROVIDED(J.E.0),TRANSFER TO(S80)-
      JJJ=J-1-
      FJP=.5*(FJP+FJA(JJJ))-

```

Fig. 6. Digital-computer program for scattering patterns of a circular dielectric cylinder having several layers.

```

S80      FJPP=.5*(FJPP+FJB(JJJ))-
          FNP=.5*(FNP+FNA(JJJ))-
          FNPP=.5*(FNPP+FNB(JJJ))-
          FJZ=FJA(J)-
          FJ1=FJB(J)-
          FNZ=FNA(J)-
          FN1=FNB(J)-
          UJ=YMM*FJZ*FNPP-YM*FJP*FN1-
          VJ=YM*FJP*FJ1-YMM*FJZ*FJPP-
          WJ=YMM*FNZ*FNPP-YM*FNP*FN1-
          XJ=YM*FNP*FJ1-YMM*FNZ*FJPP-
          AJ=A(J)-
          BJ=B(J)-
          A(J)=UJ*AJ+WJ*BJ-
          B(J)=VJ*AJ+XJ*BJ-
          UP=YMMP*FJZ*FNPP-YMP*FJP*FN1-
          VP=YMP*FJP*FJ1-YMMP*FJZ*FJPP-
          WP=YMMP*FNZ*FNPP-YMP*FNP*FN1-
          XP=YMP*FNP*FJ1-YMMP*FNZ*FJPP-
          AJ=AP(J)-
          BJ=BP(J)-
          AP(J)=UP*AJ+WP*BJ-
          BP(J)=VP*AJ+XP*BJ-
S100     CONTINUE-
S110     CONTINUE-
          DO THROUGH(S150),I=0,1,I.LE.LL-
          FI=I-
          D(I)=-B(I)/(B(I)+(.I.1.)*A(I))-
          DP(I)=-BP(I)/(BP(I)+(.I.1.)*AP(I))-
          DMAG=.ABS.D(I)-
          DMAGP=.ABS.DP(I)-
S150     WRITE OUTPUT,1,(FI,D(I),DP(I),DMAG,DMAGP)-
          WRITE OUTPUT,2-
          M=180./DPH-
          DO THROUGH(S200),I=0,1,I.LE.M-
          ES=.0-
          ESP=.0-
          FI=I-
          PHI=FI*DPH-
          PR=.01745329*PHI-
          EN=1.-
          DO THROUGH(S180),J=0,1,J.LE.LL-
          FFJ=J-
          COS=COS.(FFJ*PR)-
          ES=ES+EN*D(J)*COS-
          ESP=ESP+EN*DP(J)*COS-
S180     EN=2.-
          EMAGP=.ABS.ESP-
          ECHWP=CST*EMAGP*EMAGP-
          EMAG=.ABS.ES-
          ECHW=CST*EMAG*EMAG-
S200     WRITE OUTPUT,2,(PHI,ECHW,ECHWP)-
          END PROGRAM(FIRST)-
***      DATA
          5.0      5.0
          .1      .2      .3      .4      .5
          6.0      5.0      4.0      3.0      2.0
          1.0      1.0      1.0      1.0      1.0

```

Fig. 6. Digital-computer program for scattering patterns of a circular dielectric cylinder having several layers.